Rules for integrands involving inverse hyperbolic cosines

1. $\int u\,(a+b\,\text{ArcCosh}[c+d\,x])^n\,dx$

1: $\int (a+b\,\text{ArcCosh}[c+d\,x])^n\,dx$

Derivation: Integration by substitution

Rule:

$$\int (a+b\,\text{ArcCosh}[c+d\,x])^n\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\Big[\int (a+b\,\text{ArcCosh}[x])^n\,dx,\ x,\ c+d\,x\Big]$$

Program code:

```
Int[(a_.+b_.*ArcCosh[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[(a+b*ArcCosh[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,n},x]
```

2: $\int (e+f\,x)^m\,(a+b\,\text{ArcCosh}[c+d\,x])^n\,dx$

Derivation: Integration by substitution

Rule:

$$\int (e+f\,x)^m\,(a+b\,\text{ArcCosh}[c+d\,x])^n\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\Big[\int \left(\frac{d\,e-c\,f}{d}+\frac{f\,x}{d}\right)^m\,(a+b\,\text{ArcCosh}[x])^n\,dx,\ x,\ c+d\,x\Big]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*ArcCosh[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(a+b*ArcCosh[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

3: $\int (A + B x + C x^2)^p (a + b \operatorname{ArcCosh}[c + d x])^n \, dx$ when $B (1 - c^2) + 2 A c d == 0 \land 2 c C - B d == 0$

Derivation: Integration by substitution

Basis: If $B (1 - c^2) + 2 A c d == 0 \land 2 c C - B d == 0$, then $A + B x + C x^2 == -\frac{C}{d^2} + \frac{C}{d^2} (c + d x)^2$

Rule: If $B (1 - c^2) + 2 A c d == 0 \land 2 c C - B d == 0$, then

$$\int (A + B x + C x^2)^p (a + b \operatorname{ArcCosh}[c + d x])^n \, dx \longrightarrow \frac{1}{d} \operatorname{Subst}\left[\int \left(-\frac{C}{d^2} + \frac{C x^2}{d^2}\right)^p (a + b \operatorname{ArcCosh}[x])^n \, dx,\ x,\ c + d x\right]$$

Program code:

```
Int[(A_.+B_.*x_+C_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[(-C/d^2+C/d^2*x^2)^p*(a+b*ArcCosh[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,A,B,C,n,p},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

4: $\int (e + f x)^m (A + B x + C x^2)^p (a + b \operatorname{ArcCosh}[c + d x])^n \, dx$ when $B (1 - c^2) + 2 A c d == 0 \land 2 c C - B d == 0$

Derivation: Integration by substitution

Basis: If $B (1 - c^2) + 2 A c d == 0 \land 2 c C - B d == 0$, then $A + B x + C x^2 == -\frac{C}{d^2} + \frac{C}{d^2} (c + d x)^2$

Rule: If $B (1 - c^2) + 2 A c d == 0 \land 2 c C - B d == 0$, then

$$\int (e + f x)^m (A + B x + C x^2)^p (a + b \operatorname{ArcCosh}[c + d x])^n \, dx \longrightarrow \frac{1}{d} \operatorname{Subst}\left[\int \left(\frac{d e - c f}{d} + \frac{f x}{d}\right)^m \left(-\frac{C}{d^2} + \frac{C x^2}{d^2}\right)^p (a + b \operatorname{ArcCosh}[x])^n \, dx,\ x,\ c + d x\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(A_.+B_.*x_+C_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_+d_.*x_])^n_.,x_Symbol] :=
  1/d*Subst[Int[((d*e-c*f)/d+f*x/d)^m*(-C/d^2+C/d^2*x^2)^p*(a+b*ArcCosh[x])^n,x],x,c+d*x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n,p},x] && EqQ[B*(1-c^2)+2*A*c*d,0] && EqQ[2*c*C-B*d,0]
```

2. $\int (a + b \,\text{ArcCosh}[c + d x^2])^n \, dx$ when $c^2 == 1$

1. $\int (a + b \,\text{ArcCosh}[c + d x^2])^n \, dx$ when $c^2 == 1 \land n > 0$

1. $\int \sqrt{a + b \,\text{ArcCosh}[c + d x^2]} \, dx$ when $c^2 == 1$

1: $\int \sqrt{a + b \,\text{ArcCosh}[1 + d x^2]} \, dx$

–

Rule:

$$\int \sqrt{a + b \,\text{ArcCosh}[1 + d x^2]} \, dx \longrightarrow$$

$$\frac{2 \sqrt{a + b \,\text{ArcCosh}[1 + d x^2]} \, \text{Sinh}\left[\frac{1}{2} \text{ArcCosh}[1 + d x^2]\right]^2}{d x} -$$

$$\frac{1}{d x} \sqrt{b} \sqrt{\frac{\pi}{2}} \left(\text{Cosh}\left[\frac{a}{2 b}\right] - \text{Sinh}\left[\frac{a}{2 b}\right]\right) \text{Sinh}\left[\frac{1}{2} \text{ArcCosh}[1 + d x^2]\right] \text{Erfi}\left[\frac{1}{\sqrt{2 b}} \sqrt{a + b \,\text{ArcCosh}[1 + d x^2]}\right] +$$

$$\frac{1}{d x} \sqrt{b} \sqrt{\frac{\pi}{2}} \left(\text{Cosh}\left[\frac{a}{2 b}\right] + \text{Sinh}\left[\frac{a}{2 b}\right]\right) \text{Sinh}\left[\frac{1}{2} \text{ArcCosh}[1 + d x^2]\right] \text{Erf}\left[\frac{1}{\sqrt{2 b}} \sqrt{a + b \,\text{ArcCosh}[1 + d x^2]}\right]$$

Program code:

```
Int[Sqrt[a_.+b_.*ArcCosh[1+d_.*x_^2]],x_Symbol] :=
  2*Sqrt[a+b*ArcCosh[1+d*x^2]]*Sinh[(1/2)*ArcCosh[1+d*x^2]]^2/(d*x) -
  Sqrt[b]*Sqrt[Pi/2]*(Cosh[a/(2*b)]-Sinh[a/(2*b)])*Sinh[(1/2)*ArcCosh[1+d*x^2]]*
    Erfi[(1/Sqrt[2*b])*Sqrt[a+b*ArcCosh[1+d*x^2]]]/(d*x) +
  Sqrt[b]*Sqrt[Pi/2]*(Cosh[a/(2*b)]+Sinh[a/(2*b)])*Sinh[(1/2)*ArcCosh[1+d*x^2]]*
    Erf[(1/Sqrt[2*b])*Sqrt[a+b*ArcCosh[1+d*x^2]]]/(d*x) /;
FreeQ[{a,b,d},x]
```

2: $\int\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}\,dx$

–

Rule:

$$\int\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}\,dx\ \longrightarrow$$
$$\frac{2\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}\ \text{Cosh}\left[\frac{1}{2}\,\text{ArcCosh}\left[-1+d\,x^2\right]\right]^2}{d\,x}-$$
$$\frac{1}{d\,x}\sqrt{b}\,\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right]-\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Cosh}\left[\frac{1}{2}\,\text{ArcCosh}\left[-1+d\,x^2\right]\right]\,\text{Erfi}\left[\frac{1}{\sqrt{2\,b}}\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}\right]-$$
$$\frac{1}{d\,x}\sqrt{b}\,\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right]+\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Cosh}\left[\frac{1}{2}\,\text{ArcCosh}\left[-1+d\,x^2\right]\right]\,\text{Erf}\left[\frac{1}{\sqrt{2\,b}}\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}\right]$$

Program code:

```
Int[Sqrt[a_.+b_.*ArcCosh[-1+d_.*x_^2]],x_Symbol] :=
  2*Sqrt[a+b*ArcCosh[-1+d*x^2]]*Cosh[(1/2)*ArcCosh[-1+d*x^2]]^2/(d*x) -
  Sqrt[b]*Sqrt[Pi/2]*(Cosh[a/(2*b)]-Sinh[a/(2*b)])*Cosh[(1/2)*ArcCosh[-1+d*x^2]]*
    Erfi[(1/Sqrt[2*b])*Sqrt[a+b*ArcCosh[-1+d*x^2]]]/(d*x) -
  Sqrt[b]*Sqrt[Pi/2]*(Cosh[a/(2*b)]+Sinh[a/(2*b)])*Cosh[(1/2)*ArcCosh[-1+d*x^2]]*
    Erf[(1/Sqrt[2*b])*Sqrt[a+b*ArcCosh[-1+d*x^2]]]/(d*x) /;
FreeQ[{a,b,d},x]
```

2: $\int\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^n\,dx$ when $c^2==1\land n>1$

Derivation: Integration by parts and piecewise constant extraction both twice!

Basis: $\partial_x\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^n==\frac{2\,b\,d\,n\,x\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^{n-1}}{\sqrt{-1+c+d\,x^2}\,\sqrt{1+c+d\,x^2}}$

Basis: If $c^2==1$, then $\partial_x\frac{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}{\sqrt{-1+c+d\,x^2}\,\sqrt{1+c+d\,x^2}}==0$

Basis: $\frac{x^2}{\sqrt{2\,c\,d\,x^2+d^2\,x^4}} == \partial_x \frac{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}{d^2\,x}$

Rule: If $c^2 == 1 \land n > 1$, then

$$\int \left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^n dx \longrightarrow x\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^n - 2\,b\,d\,n \int \frac{x^2\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^{n-1}}{\sqrt{-1+c+d\,x^2}\,\sqrt{1+c+d\,x^2}}\,dx$$

$$\longrightarrow x\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^n - \frac{2\,b\,d\,n\,\sqrt{2\,c\,d\,x^2+d^2\,x^4}}{\sqrt{-1+c+d\,x^2}\,\sqrt{1+c+d\,x^2}} \int \frac{x^2\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^{n-1}}{\sqrt{2\,c\,d\,x^2+d^2\,x^4}}\,dx$$

$$\longrightarrow x\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^n - \frac{2\,b\,n\left(2\,c\,d\,x^2+d^2\,x^4\right)\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^{n-1}}{d\,x\,\sqrt{-1+c+d\,x^2}\,\sqrt{1+c+d\,x^2}} + 4\,b^2\,n\,(n-1)\int \left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^{n-2} dx$$

Program code:

```
Int[(a_.+b_.*ArcCosh[c_+d_.*x_^2])^n_,x_Symbol] :=
  x*(a+b*ArcCosh[c+d*x^2])^n -
  2*b*n*(2*c*d*x^2+d^2*x^4)*(a+b*ArcCosh[c+d*x^2])^(n-1)/(d*x*Sqrt[-1+c+d*x^2]*Sqrt[1+c+d*x^2]) +
  4*b^2*n*(n-1)*Int[(a+b*ArcCosh[c+d*x^2])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1] && GtQ[n,1]
```

2. $\int (a + b\,\text{ArcCosh}[c + d\,x^2])^n\,dx$ when $c^2 == 1 \land n < 0$

   1. $\int \frac{1}{a + b\,\text{ArcCosh}[c + d\,x^2]}\,dx$ when $c^2 == 1$

      1: $\int \frac{1}{a + b\,\text{ArcCosh}[1 + d\,x^2]}\,dx$

Rule:

$$\int \frac{1}{a + b\,\text{ArcCosh}[1 + d\,x^2]}\,dx \longrightarrow$$
$$\frac{x\,\text{Cosh}\left[\frac{a}{2\,b}\right]\,\text{CoshIntegral}\left[\frac{1}{2\,b}\left(a + b\,\text{ArcCosh}[1 + d\,x^2]\right)\right]}{\sqrt{2}\,b\,\sqrt{d\,x^2}} - \frac{x\,\text{Sinh}\left[\frac{a}{2\,b}\right]\,\text{SinhIntegral}\left[\frac{1}{2\,b}\left(a + b\,\text{ArcCosh}[1 + d\,x^2]\right)\right]}{\sqrt{2}\,b\,\sqrt{d\,x^2}}$$

Program code:

```
Int[1/(a_.+b_.*ArcCosh[1+d_.*x_^2]),x_Symbol] :=
  x*Cosh[a/(2*b)]*CoshIntegral[(a+b*ArcCosh[1+d*x^2])/(2*b)]/(Sqrt[2]*b*Sqrt[d*x^2]) -
  x*Sinh[a/(2*b)]*SinhIntegral[(a+b*ArcCosh[1+d*x^2])/(2*b)]/(Sqrt[2]*b*Sqrt[d*x^2]) /;
FreeQ[{a,b,d},x]
```

2: $\int \frac{1}{a + b \operatorname{ArcCosh}\left[-1 + d x^2\right]} \, dx$

Rule:

$$\int \frac{1}{a + b \operatorname{ArcCosh}\left[-1 + d x^2\right]} \, dx \longrightarrow$$
$$-\frac{x \operatorname{Sinh}\left[\frac{a}{2 b}\right] \operatorname{CoshIntegral}\left[\frac{1}{2 b} \left(a + b \operatorname{ArcCosh}\left[-1 + d x^2\right]\right)\right]}{\sqrt{2}\, b \sqrt{d x^2}} + \frac{x \operatorname{Cosh}\left[\frac{a}{2 b}\right] \operatorname{SinhIntegral}\left[\frac{1}{2 b} \left(a + b \operatorname{ArcCosh}\left[-1 + d x^2\right]\right)\right]}{\sqrt{2}\, b \sqrt{d x^2}}$$

Program code:

```
Int[1/(a_.+b_.*ArcCosh[-1+d_.*x_^2]),x_Symbol] :=
  -x*Sinh[a/(2*b)]*CoshIntegral[(a+b*ArcCosh[-1+d*x^2])/(2*b)]/(Sqrt[2]*b*Sqrt[d*x^2]) +
  x*Cosh[a/(2*b)]*SinhIntegral[(a+b*ArcCosh[-1+d*x^2])/(2*b)]/(Sqrt[2]*b*Sqrt[d*x^2]) /;
FreeQ[{a,b,d},x]
```

2. $\int \frac{1}{\sqrt{a+b\,\text{ArcCosh}\left[c+d\,x^2\right]}}\,dx$ when $c^2 == 1$

1: $\int \frac{1}{\sqrt{a+b\,\text{ArcCosh}\left[1+d\,x^2\right]}}\,dx$

Rule:

$$\int \frac{1}{\sqrt{a+b\,\text{ArcCosh}\left[1+d\,x^2\right]}}\,dx \to$$

$$\frac{1}{\sqrt{b}\;d\,x}\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right]-\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Sinh}\left[\frac{1}{2}\,\text{ArcCosh}\left[1+d\,x^2\right]\right]\,\text{Erfi}\left[\frac{1}{\sqrt{2\,b}}\,\sqrt{a+b\,\text{ArcCosh}\left[1+d\,x^2\right]}\,\right]+$$
$$\frac{1}{\sqrt{b}\;d\,x}\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right]+\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Sinh}\left[\frac{1}{2}\,\text{ArcCosh}\left[1+d\,x^2\right]\right]\,\text{Erf}\left[\frac{1}{\sqrt{2\,b}}\,\sqrt{a+b\,\text{ArcCosh}\left[1+d\,x^2\right]}\,\right]$$

Program code:

```
Int[1/Sqrt[a_.+b_.*ArcCosh[1+d_.*x_^2]],x_Symbol] :=
  Sqrt[Pi/2]*(Cosh[a/(2*b)]-Sinh[a/(2*b)])*Sinh[ArcCosh[1+d*x^2]/2]*Erfi[Sqrt[a+b*ArcCosh[1+d*x^2]]/Sqrt[2*b]]/(Sqrt[b]*d*x) +
  Sqrt[Pi/2]*(Cosh[a/(2*b)]+Sinh[a/(2*b)])*Sinh[ArcCosh[1+d*x^2]/2]*Erf[Sqrt[a+b*ArcCosh[1+d*x^2]]/Sqrt[2*b]]/(Sqrt[b]*d*x) /;
FreeQ[{a,b,d},x]
```

2: $\int \frac{1}{\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}}\,dx$

Rule:

$$\int \frac{1}{\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}}\,dx \to$$

$$\frac{1}{\sqrt{b}\;d\,x}\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right]-\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Cosh}\left[\frac{1}{2}\,\text{ArcCosh}\left[-1+d\,x^2\right]\right]\,\text{Erfi}\left[\frac{1}{\sqrt{2\,b}}\,\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}\,\right]-$$

$$\frac{1}{\sqrt{b}\ d\,x}\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right]+\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Cosh}\left[\frac{1}{2}\,\text{ArcCosh}\left[-1+d\,x^2\right]\right]\text{Erf}\left[\frac{1}{\sqrt{2\,b}}\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}\right]$$

Program code:

```
Int[1/Sqrt[a_.+b_.*ArcCosh[-1+d_.*x_^2]],x_Symbol] :=
  Sqrt[Pi/2]*(Cosh[a/(2*b)]-Sinh[a/(2*b)])*Cosh[ArcCosh[-1+d*x^2]/2]*Erfi[Sqrt[a+b*ArcCosh[-1+d*x^2]]/Sqrt[2*b]]/(Sqrt[b]*d*x) -
  Sqrt[Pi/2]*(Cosh[a/(2*b)]+Sinh[a/(2*b)])*Cosh[ArcCosh[-1+d*x^2]/2]*Erf[Sqrt[a+b*ArcCosh[-1+d*x^2]]/Sqrt[2*b]]/(Sqrt[b]*d*x) /;
FreeQ[{a,b,d},x]
```

3. $\int (a + b\,\text{ArcCosh}[c + d\,x^2])^n\,dx$ when $c^2 == 1 \land n < -1$

   1. $\int \frac{1}{(a + b\,\text{ArcCosh}[c + d\,x^2])^{3/2}}\,dx$ when $c^2 == 1$

      1: $\int \frac{1}{(a + b\,\text{ArcCosh}[1 + d\,x^2])^{3/2}}\,dx$

Derivation: Integration by parts

Basis: $-\frac{b\,d\,x}{\sqrt{d\,x^2}\,\sqrt{2+d\,x^2}\,(a+b\,\text{ArcCosh}[1+d\,x^2])^{3/2}} == \partial_x \frac{1}{\sqrt{a+b\,\text{ArcCosh}[1+d\,x^2]}}$

Rule:

$$\int \frac{1}{(a + b\,\text{ArcCosh}[1 + d\,x^2])^{3/2}}\,dx \longrightarrow -\frac{\sqrt{d\,x^2}\,\sqrt{2 + d\,x^2}}{b\,d\,x\,\sqrt{a + b\,\text{ArcCosh}[1 + d\,x^2]}} + \frac{d}{b}\int \frac{x^2}{\sqrt{d\,x^2}\,\sqrt{2 + d\,x^2}\,\sqrt{a + b\,\text{ArcCosh}[1 + d\,x^2]}}\,dx$$

$$\longrightarrow -\frac{\sqrt{d\,x^2}\,\sqrt{2 + d\,x^2}}{b\,d\,x\,\sqrt{a + b\,\text{ArcCosh}[1 + d\,x^2]}} +$$

$$\frac{1}{b^{3/2}\,d\,x}\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right] - \text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Sinh}\left[\frac{1}{2}\,\text{ArcCosh}[1 + d\,x^2]\right]\,\text{Erfi}\left[\frac{1}{\sqrt{2\,b}}\,\sqrt{a + b\,\text{ArcCosh}[1 + d\,x^2]}\,\right] -$$

$$\frac{1}{b^{3/2}\,d\,x}\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right] + \text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Sinh}\left[\frac{1}{2}\,\text{ArcCosh}[1 + d\,x^2]\right]\,\text{Erf}\left[\frac{1}{\sqrt{2\,b}}\,\sqrt{a + b\,\text{ArcCosh}[1 + d\,x^2]}\,\right]$$

Program code:

```
Int[1/(a_.+b_.*ArcCosh[1+d_.*x_^2])^(3/2),x_Symbol] :=
  -Sqrt[d*x^2]*Sqrt[2+d*x^2]/(b*d*x*Sqrt[a+b*ArcCosh[1+d*x^2]]) +
  Sqrt[Pi/2]*(Cosh[a/(2*b)]-Sinh[a/(2*b)])*Sinh[ArcCosh[1+d*x^2]/2]*Erfi[Sqrt[a+b*ArcCosh[1+d*x^2]]/Sqrt[2*b]]/(b^(3/2)*d*x)-
  Sqrt[Pi/2]*(Cosh[a/(2*b)]+Sinh[a/(2*b)])*Sinh[ArcCosh[1+d*x^2]/2]*Erf[Sqrt[a+b*ArcCosh[1+d*x^2]]/Sqrt[2*b]]/(b^(3/2)*d*x) /;
FreeQ[{a,b,d},x]
```

2: $\int \frac{1}{\left(a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]\right)^{3/2}}\,dx$

- Derivation: Integration by parts

- Basis: $-\frac{b\,d\,x}{\sqrt{d\,x^2}\,\sqrt{-2+d\,x^2}\,\left(a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]\right)^{3/2}} = \partial_x \frac{1}{\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}}$

- Rule:

$$\int \frac{1}{\left(a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]\right)^{3/2}}\,dx \to -\frac{\sqrt{d\,x^2}\,\sqrt{-2+d\,x^2}}{b\,d\,x\,\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}}+\frac{d}{b}\int \frac{x^2}{\sqrt{d\,x^2}\,\sqrt{-2+d\,x^2}\,\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}}\,dx$$

$$\to -\frac{\sqrt{d\,x^2}\,\sqrt{-2+d\,x^2}}{b\,d\,x\,\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}}+$$

$$\frac{1}{b^{3/2}\,d\,x}\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right]-\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Cosh}\left[\frac{1}{2}\,\text{ArcCosh}\left[-1+d\,x^2\right]\right]\,\text{Erfi}\left[\frac{1}{\sqrt{2\,b}}\,\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}\right]+$$

$$\frac{1}{b^{3/2}\,d\,x}\sqrt{\frac{\pi}{2}}\left(\text{Cosh}\left[\frac{a}{2\,b}\right]+\text{Sinh}\left[\frac{a}{2\,b}\right]\right)\text{Cosh}\left[\frac{1}{2}\,\text{ArcCosh}\left[-1+d\,x^2\right]\right]\,\text{Erf}\left[\frac{1}{\sqrt{2\,b}}\,\sqrt{a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]}\right]$$

- Program code:

```
Int[1/(a_.+b_.*ArcCosh[-1+d_.*x_^2])^(3/2),x_Symbol] :=
  -Sqrt[d*x^2]*Sqrt[-2+d*x^2]/(b*d*x*Sqrt[a+b*ArcCosh[-1+d*x^2]]) +
  Sqrt[Pi/2]*(Cosh[a/(2*b)]-Sinh[a/(2*b)])*Cosh[ArcCosh[-1+d*x^2]/2]*Erfi[Sqrt[a+b*ArcCosh[-1+d*x^2]]/Sqrt[2*b]]/(b^(3/2)*d*x) +
  Sqrt[Pi/2]*(Cosh[a/(2*b)]+Sinh[a/(2*b)])*Cosh[ArcCosh[-1+d*x^2]/2]*Erf[Sqrt[a+b*ArcCosh[-1+d*x^2]]/Sqrt[2*b]]/(b^(3/2)*d*x) /;
FreeQ[{a,b,d},x]
```

2. $\int \frac{1}{\left(a+b\,\text{ArcCosh}\left[c+d\,x^2\right]\right)^2}\,dx$ when $c^2 == 1$

1: $\int \frac{1}{\left(a+b\,\text{ArcCosh}\left[1+d\,x^2\right]\right)^2}\,dx$

- Rule:

$$\int \frac{1}{\left(a+b\,\text{ArcCosh}\left[1+d\,x^2\right]\right)^2}\,dx \longrightarrow$$
$$-\frac{\sqrt{d\,x^2}\,\sqrt{2+d\,x^2}}{2\,b\,d\,x\,\left(a+b\,\text{ArcCosh}\left[1+d\,x^2\right]\right)}-\frac{x\,\text{Sinh}\left[\frac{a}{2\,b}\right]\,\text{CoshIntegral}\left[\frac{1}{2\,b}\left(a+b\,\text{ArcCosh}\left[1+d\,x^2\right]\right)\right]}{2\,\sqrt{2}\,b^2\,\sqrt{d\,x^2}}+\frac{x\,\text{Cosh}\left[\frac{a}{2\,b}\right]\,\text{SinhIntegral}\left[\frac{1}{2\,b}\left(a+b\,\text{ArcCosh}\left[1+d\,x^2\right]\right)\right]}{2\,\sqrt{2}\,b^2\,\sqrt{d\,x^2}}$$

Program code:

```
Int[1/(a_.+b_.*ArcCosh[1+d_.*x_^2])^2,x_Symbol] :=
  -Sqrt[d*x^2]*Sqrt[2+d*x^2]/(2*b*d*x*(a+b*ArcCosh[1+d*x^2])) -
  x*Sinh[a/(2*b)]*CoshIntegral[(a+b*ArcCosh[1+d*x^2])/(2*b)]/(2*Sqrt[2]*b^2*Sqrt[d*x^2]) +
  x*Cosh[a/(2*b)]*SinhIntegral[(a+b*ArcCosh[1+d*x^2])/(2*b)]/(2*Sqrt[2]*b^2*Sqrt[d*x^2]) /;
FreeQ[{a,b,d},x]
```

2: $\int \frac{1}{\left(a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]\right)^2}\,dx$

Rule:

$$\int \frac{1}{\left(a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]\right)^2}\,dx \longrightarrow$$
$$-\frac{\sqrt{d\,x^2}\,\sqrt{-2+d\,x^2}}{2\,b\,d\,x\,\left(a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]\right)}+\frac{x\,\text{Cosh}\left[\frac{a}{2\,b}\right]\,\text{CoshIntegral}\left[\frac{1}{2\,b}\left(a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]\right)\right]}{2\,\sqrt{2}\,b^2\,\sqrt{d\,x^2}}-\frac{x\,\text{Sinh}\left[\frac{a}{2\,b}\right]\,\text{SinhIntegral}\left[\frac{1}{2\,b}\left(a+b\,\text{ArcCosh}\left[-1+d\,x^2\right]\right)\right]}{2\,\sqrt{2}\,b^2\,\sqrt{d\,x^2}}$$

Program code:

```
Int[1/(a_.+b_.*ArcCosh[-1+d_.*x_^2])^2,x_Symbol] :=
  -Sqrt[d*x^2]*Sqrt[-2+d*x^2]/(2*b*d*x*(a+b*ArcCosh[-1+d*x^2])) +
  x*Cosh[a/(2*b)]*CoshIntegral[(a+b*ArcCosh[-1+d*x^2])/(2*b)]/(2*Sqrt[2]*b^2*Sqrt[d*x^2]) -
  x*Sinh[a/(2*b)]*SinhIntegral[(a+b*ArcCosh[-1+d*x^2])/(2*b)]/(2*Sqrt[2]*b^2*Sqrt[d*x^2]) /;
FreeQ[{a,b,d},x]
```

3: $\int (a + b\,\text{ArcCosh}[c + d\,x^2])^n\, dx$ when $c^2 == 1 \land n < -1 \land n \neq -2$

Derivation: Inverted integration by parts and piecewise constant extraction both twice!

Rule: If $c^2 == 1 \land n < -1 \land n \neq -2$, then

$$\int (a + b\,\text{ArcCosh}[c + d\,x^2])^n\, dx \longrightarrow$$

$$-\frac{x\,(a + b\,\text{ArcCosh}[c + d\,x^2])^{n+2}}{4\,b^2\,(n+1)\,(n+2)} + \frac{(2\,c\,x^2 + d\,x^4)\,(a + b\,\text{ArcCosh}[c + d\,x^2])^{n+1}}{2\,b\,(n+1)\,x\,\sqrt{-1 + c + d\,x^2}\,\sqrt{1 + c + d\,x^2}} + \frac{1}{4\,b^2\,(n+1)\,(n+2)} \int (a + b\,\text{ArcCosh}[c + d\,x^2])^{n+2}\, dx$$

Program code:

```
Int[(a_.+b_.*ArcCosh[c_+d_.*x_^2])^n_,x_Symbol] :=
  -x*(a+b*ArcCosh[c+d*x^2])^(n+2)/(4*b^2*(n+1)*(n+2)) +
  (2*c*x^2 +d*x^4)*(a+b*ArcCosh[c+d*x^2])^(n+1)/(2*b*(n+1)*x*Sqrt[-1+c+d*x^2]*Sqrt[1+c+d*x^2]) +
  1/(4*b^2*(n+1)*(n+2))*Int[(a+b*ArcCosh[c+d*x^2])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && EqQ[c^2,1] && LtQ[n,-1] && NeQ[n,-2]
```

3: $\int \frac{\text{ArcCosh}[a\,x^p]^n}{x}\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $\frac{\text{ArcCosh}[a\,x^p]^n}{x} == \frac{1}{p}\,\text{Subst}[x^n\,\text{Tanh}[x],\,x,\,\text{ArcCosh}[a\,x^p]]\;\partial_x\,\text{ArcCosh}[a\,x^p]$

Rule: If $n \in \mathbb{Z}^+$, then

$$\int \frac{\text{ArcCosh}[a\,x^p]^n}{x}\,dx \longrightarrow \frac{1}{p}\,\text{Subst}\Big[\int x^n\,\text{Tanh}[x]\,dx,\,x,\,\text{ArcCosh}[a\,x^p]\Big]$$

Program code:

```
Int[ArcCosh[a_.*x_^p_]^n_./x_,x_Symbol] :=
  1/p*Subst[Int[x^n*Tanh[x],x],x,ArcCosh[a*x^p]] /;
FreeQ[{a,p},x] && IGtQ[n,0]
```

4: $\int u \operatorname{ArcCosh}\left[\frac{c}{a+b x^n}\right]^m dx$

Derivation: Algebraic simplification

Basis: $\operatorname{ArcCosh}[z] == \operatorname{ArcSech}\left[\frac{1}{z}\right]$

Rule:

$$\int u \operatorname{ArcCosh}\left[\frac{c}{a+b x^n}\right]^m dx \rightarrow \int u \operatorname{ArcSech}\left[\frac{a}{c}+\frac{b x^n}{c}\right]^m dx$$

Program code:

```
Int[u_.*ArcCosh[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcSech[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

5: $\int \frac{\operatorname{ArcCosh}\left[\sqrt{1+b x^2}\right]^n}{\sqrt{1+b x^2}} dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{\sqrt{-1+\sqrt{1+b x^2}}\ \sqrt{1+\sqrt{1+b x^2}}}{x} == 0$

Basis: $\frac{x \operatorname{ArcCosh}\left[\sqrt{1+b x^2}\right]^n}{\sqrt{-1+\sqrt{1+b x^2}}\ \sqrt{1+\sqrt{1+b x^2}}\ \sqrt{1+b x^2}} == \frac{1}{b} \operatorname{Subst}\left[\frac{\operatorname{ArcCosh}[x]^n}{\sqrt{-1+x}\ \sqrt{1+x}}, x, \sqrt{1+b x^2}\right] \partial_x \sqrt{1+b x^2}$

Rule:

$$\int \frac{\operatorname{ArcCosh}\left[\sqrt{1+b x^2}\right]^n}{\sqrt{1+b x^2}} dx \rightarrow \frac{\sqrt{-1+\sqrt{1+b x^2}}\ \sqrt{1+\sqrt{1+b x^2}}}{x} \int \frac{x \operatorname{ArcCosh}\left[\sqrt{1+b x^2}\right]^n}{\sqrt{-1+\sqrt{1+b x^2}}\ \sqrt{1+\sqrt{1+b x^2}}\ \sqrt{1+b x^2}} dx$$

$$\longrightarrow \frac{\sqrt{-1+\sqrt{1+b\,x^2}}\,\sqrt{1+\sqrt{1+b\,x^2}}}{b\,x}\,\text{Subst}\Big[\int\frac{\text{ArcCosh}[x]^n}{\sqrt{-1+x}\,\sqrt{1+x}}\,dx,\,x,\,\sqrt{1+b\,x^2}\Big]$$

– Program code:

```
Int[ArcCosh[Sqrt[1+b_.*x_^2]]^n_./Sqrt[1+b_.*x_^2],x_Symbol] :=
  Sqrt[-1+Sqrt[1+b*x^2]]*Sqrt[1+Sqrt[1+b*x^2]]/(b*x)*Subst[Int[ArcCosh[x]^n/(Sqrt[-1+x]*Sqrt[1+x]),x],x,Sqrt[1+b*x^2]] /;
FreeQ[{b,n},x]
```

6. $\int u\, f^{c\,\text{ArcCosh}[a+b\,x]^n}\,dx$ when $n\in\mathbb{Z}^+$

1: $\int f^{c\,\text{ArcCosh}[a+b\,x]^n}\,dx$ when $n\in\mathbb{Z}^+$

– Derivation: Integration by substitution

– Basis: $F[\text{ArcCosh}[a+b\,x]] = \frac{1}{b}\,\text{Subst}[F[x]\,\text{Sinh}[x],\,x,\,\text{ArcCosh}[a+b\,x]]\,\partial_x\text{ArcCosh}[a+b\,x]$

– Rule: If $n\in\mathbb{Z}^+$, then

$$\int f^{c\,\text{ArcCosh}[a+b\,x]^n}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\Big[\int f^{c\,x^n}\,\text{Sinh}[x]\,dx,\,x,\,\text{ArcCosh}[a+b\,x]\Big]$$

– Program code:

```
Int[f_^(c_.*ArcCosh[a_.+b_.*x_]^n_.),x_Symbol] :=
  1/b*Subst[Int[f^(c*x^n)*Sinh[x],x],x,ArcCosh[a+b*x]] /;
FreeQ[{a,b,c,f},x] && IGtQ[n,0]
```

2: $\int x^m f^{c\,\text{ArcCosh}[a+b\,x]^n}\,dx$ when $(m \mid n) \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x, \text{ArcCosh}[a+b\,x]] = \frac{1}{b}\,\text{Subst}\left[F\left[-\frac{a}{b}+\frac{\text{Cosh}[x]}{b}, x\right]\text{Sinh}[x], x, \text{ArcCosh}[a+b\,x]\right]\partial_x \text{ArcCosh}[a+b\,x]$

Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int x^m f^{c\,\text{ArcCosh}[a+b\,x]^n}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int\left(-\frac{a}{b}+\frac{\text{Cosh}[x]}{b}\right)^m f^{c\,x^n}\,\text{Sinh}[x]\,dx, x, \text{ArcCosh}[a+b\,x]\right]$$

Program code:

```
Int[x_^m_.*f_^(c_.*ArcCosh[a_.+b_.*x_]^n_.),x_Symbol] :=
  1/b*Subst[Int[(-a/b+Cosh[x]/b)^m*f^(c*x^n)*Sinh[x],x],x,ArcCosh[a+b*x]] /;
FreeQ[{a,b,c,f},x] && IGtQ[m,0] && IGtQ[n,0]
```

7. $\int v\,(a+b\,\text{ArcCosh}[u])\,dx$ when u is free of inverse functions

1: $\int \text{ArcCosh}[u]\,dx$ when u is free of inverse functions

Derivation: Integration by parts

Basis: $\partial_x \text{ArcCosh}[f[x]] == \frac{\partial_x f[x]}{\sqrt{-1+f[x]}\,\sqrt{1+f[x]}}$

Rule: If u is free of inverse functions, then

$$\int \text{ArcCosh}[u]\,dx \longrightarrow x\,\text{ArcCosh}[u] - \int \frac{x\,\partial_x u}{\sqrt{-1+u}\,\sqrt{1+u}}\,dx$$

Program code:

```
Int[ArcCosh[u_],x_Symbol] :=
  x*ArcCosh[u] -
  Int[SimplifyIntegrand[x*D[u,x]/(Sqrt[-1+u]*Sqrt[1+u]),x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialQ[u,x]]
```

2: $\int (c + d\,x)^m\,(a + b\,\text{ArcCosh}[u])\,dx$ when $m \neq -1 \land u$ is free of inverse functions

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x \text{ArcCosh}[f[x]] == \frac{\partial_x f[x]}{\sqrt{-1+f[x]}\,\sqrt{1+f[x]}}$

Rule: If $m \neq -1 \land u$ is free of inverse functions, then

$$\int (c + d\,x)^m\,(a + b\,\text{ArcCosh}[u])\,dx \longrightarrow \frac{(c + d\,x)^{m+1}\,(a + b\,\text{ArcCosh}[u])}{d\,(m + 1)} - \frac{b}{d\,(m + 1)} \int \frac{(c + d\,x)^{m+1}\,\partial_x u}{\sqrt{-1 + u}\,\sqrt{1 + u}}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*(a_.+b_.*ArcCosh[u_]),x_Symbol] :=
  (c+d*x)^(m+1)*(a+b*ArcCosh[u])/(d*(m+1)) -
  b/(d*(m+1))*Int[SimplifyIntegrand[(c+d*x)^(m+1)*D[u,x]/(Sqrt[-1+u]*Sqrt[1+u]),x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1] && InverseFunctionFreeQ[u,x] && Not[FunctionOfQ[(c+d*x)^(m+1),u,x]] && Not[FunctionOfExponentialQ[u,x]]
```

3: $\int v\,(a+b\,\text{ArcCosh}[u])\,dx$ when u and $\int v\,dx$ are free of inverse functions

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x \text{ArcCosh}[f[x]] == \frac{\partial_x f[x]}{\sqrt{-1+f[x]}\,\sqrt{1+f[x]}}$

Rule: If u is free of inverse functions, let $w == \int v\,dx$, if w is free of inverse functions, then

$$\int v\,(a+b\,\text{ArcCosh}[u])\,dx \longrightarrow w\,(a+b\,\text{ArcCosh}[u])-b\int\frac{w\,\partial_x u}{\sqrt{-1+u}\,\sqrt{1+u}}\,dx$$

Program code:

```
Int[v_*(a_.+b_.*ArcCosh[u_]),x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[(a+b*ArcCosh[u]),w,x] - b*Int[SimplifyIntegrand[w*D[u,x]/(Sqrt[-1+u]*Sqrt[1+u]),x],x] /;
 InverseFunctionFreeQ[w,x]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x] && Not[MatchQ[v, (c_.+d_.*x)^m_. /; FreeQ[{c,d,m},x]]]
```

8. $\int u\, e^{n\,\text{ArcCosh}[P_x]}\, dx$

1: $\int e^{n\,\text{ArcCosh}[P_x]}\, dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcCosh}[z]} == \left(z + \sqrt{-1+z}\,\sqrt{1+z}\right)^n$

Basis: If $n \in \mathbb{Z}$, $e^{n\,\text{ArcCosh}[z]} == \left(z + \sqrt{\frac{-1+z}{1+z}} + z\sqrt{\frac{-1+z}{1+z}}\right)^n$

Rule: If $n \in \mathbb{Z}$, then

$$\int e^{n\,\text{ArcCosh}[P_x]}\, dx \longrightarrow \int \left(P_x + \sqrt{-1+P_x}\,\sqrt{1+P_x}\right)^n dx$$

Program code:

```
Int[E^(n_.*ArcCosh[u_]), x_Symbol] :=
  Int[(u+Sqrt[-1+u]*Sqrt[1+u])^n,x] /;
IntegerQ[n] && PolyQ[u,x]
```

2: $\int x^m e^{n \operatorname{ArcCosh}[P_x]} \, dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n \operatorname{ArcCosh}[z]} == \left(z + \sqrt{-1+z} \sqrt{1+z}\right)^n$

Rule: If $n \in \mathbb{Z}$, then

$$\int x^m e^{n \operatorname{ArcCosh}[P_x]} \, dx \longrightarrow \int x^m \left(P_x + \sqrt{-1+P_x} \sqrt{1+P_x}\right)^n dx$$

Program code:

```
Int[x_^m_.*E^(n_.*ArcCosh[u_]), x_Symbol] :=
  Int[x^m*(u+Sqrt[-1+u]*Sqrt[1+u])^n,x] /;
RationalQ[m] && IntegerQ[n] && PolyQ[u,x]
```